PERMUTATION GROUPS IN MAPLE

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog and include the title or URL of this post in your comment.

Post date: 2 January 2026.

Maple has a large library of methods that deal with permutation groups, also known as symmetric groups. All these methods are in the GroupTheory library, so this must be included in any worksheet.

PERMUTATIONS

For single permutations (outside of a group) we can define a permutation as

```
with(GroupTheory):
P3 := Perm([[3, 4, 1]]);
                        P3 := (1, 3, 4)
```

This defines the permutation that maps $3\to4\to1\to3$. Note that the argument of the `Perm` function is a list of lists, so we need double brackets. Maple's response gives the permutation in the order with the lowest number first.

For a permutation defined as a product of disjoint cycles, as in

$$\phi=\left(\begin{array}{cc}1 & 3\end{array}\right)\left(\begin{array}{ccc}2 & 4 & 5\end{array}\right) \tag{1}$$

we list each cycle as a list in the argument to `Perm`:

```
P32 := Perm([[1, 3], [2, 4, 5]]);
                    P32 := (1,3)(2,4,5)
```

This maps $1\to3\to1$ and $2\to4\to5\to2$.

To find the mapping of one number onto another, for example, the mapping of 4, we can say:

```
P32[4];
                              5
```

This shows that 4 is mapped into 5.

To find the product of two permutations, we can use two commands:

```
PermProduct(P3, P32);
                          (2,4,3,5)
P3 . P32;
```

```
                                         (2,4,3,5)
P32 . P3;
                                         (1,4,5,2)
```

Note that, unlike standard notation in textbooks, the product of two permutations in Maple applies the left permutation first, then the right permutation. In this example, applying P3 followed by P32 maps $1 \to 3 \to 1$ (so 1 is not moved and doesn't appear in the product), then $3 \to 4 \to 5$, $4 \to 1 \to 3$ and $5 \to 2$. Note that products are not commutative, as `P3.P32`$\neq$`P32.P3`.

Powers of permutations can be found using exponential notation:

```
P3^2;
                                         (1,4,3)
P3^16;
                                         (1,3,4)
```

PERMUTATION GROUPS

The permutation group of order n can be specified using the `Symm` function. The result is a group data structure which contains a variety of information. The elements of a group can be listed using the `Elements` function. For the symmetric group of 3 objects, we have:

```
S3 := Symm(3);
Elements(S3);
                 {(1,3,2), (1,3), (2,3), (), (1,2,3), (1,2)}
```

We can find all the subgroups of a group using `SubgroupLattice`. For example

```
S3Sub := SubgroupLattice(S3);
```

To extract individual subgroups, we can do the following:

```
S3SubList := convert(S3Sub, 'list');
      S3SubList := [<>, <(2,3)>, <(1,2)>, <(1,3)>, <(1,3,2)>,
                     <(1,3,2),(1,3),(2,3),(1,2)>]
```

The list gives each subgroup in the form of a group generated by a set (hence the $< \ldots >$ notation). To get the actual elements of one of these subgroups, we can use `Elements` as above. For example:

```
S3SubList[3];
                                         <(1,2)>
Elements(S3SubList[3]);
                                       {(1,2), ()}
```

The set is $\{(1,2)\}$ which generates the group $\{(1,2),e\}$ where $e=()$ is the identity, since $(1,2)$ is its own inverse. For the last subgroup listed in `S3SubList`, we have

```
S3SubList[6];
                        <(1,3,2),(1,3),(2,3),(1,2))>
Elements(S3SubList[6]);
              {(1,3,2), (1,3), (2,3), (), (1,2,3), (1,2)}
```

In this case the subgroup is the entire group `S3`.

While these commands may seem overkill for a small group such as `S3`, they prove much more useful for larger groups. For example, the symmetric group $\mathfrak{S}_4$, the permutation on 4 objects, has a total of 30 subgroups, so isn't practical to work with by hand. $\mathfrak{S}_5$ has 156 subgroups, and $\mathfrak{S}_6$ has 1431, although in this case, Maple warns us that the group is not solvable and some subgroups may be missing.

GENERATING GROUPS

Although the total number of subgroups rapidly increases with the size of the permutation group, we can find a given subgroup from its generator. We can specify a permutation group by giving its generator in the `PermutationGroup` command:

```
g1 := PermutationGroup({[[1, 2]], [[1, 2, 3], [4, 5]]});
                    g1 := <(1,2),(1,2,3),(4,5)>
Elements(g1);
                       {(1,3,2)(4,5), (1,3,2), (1,3), (1,3)(4,5),
                            (2,3), (2,3)(4,5), (4,5), (),
                       (1,2,3), (1,2)(4,5), (1,2), (1,2,3)(4,5)}
```

Here, the group `g1` is a subgroup of $\mathfrak{S}_5$ generated by $<(1,2),(1,2,3),(4,5)>$ and has the elements shown. The order (number of elements) of a group can be found with `GroupOrder`:

```
GroupOrder(g1);
                              12
```

PINGBACKS

Pingback: Cosets of subgroups
Pingback: Cosets of subgroups in Maple
Pingback: Normal subgroups in Maple